arXiv:2101.11320v3 [cs.PL] 4 Mar 2021

TUTORIAL ON IMPLEMENTING HOARE LOGIC FOR IMPERATIVE PROGRAMS IN HASKELL

A PREPRINT

Boro Sitnikovski
Skopje, North Macedonia
buritomath@gmail.com

March 8, 2021

ABSTRACT

Using the programming language Haskell, we introduce an implementation for a simple imperative language that can evaluate arithmetic and boolean expressions; step by step, we will expand the implementation starting from arithmetic expressions, to boolean expressions and then to imperative programs. As we expand the language, we will show several evaluation strategies, considering the normalization property and how it affects an implementation. Finally, we will provide a bottom-up implementation of Hoare's logic which will allow us to deduce facts about programs without the need for a full evaluation.

Keywords Imperative languages · Functional languages · Hoare logic · Formal verification · Haskell

1 Introduction

Nowadays, imperative programming languages run the world, to list a few popular ones: C, Python, JavaScript, PHP. Code is usually written in these languages by using an imperative style; that is, the computer is being told ("commanded") what to do and how to do it specifically. The mathematical language is very unlike this, it is more declarative rather than imperative, that is, it doesn't care about the how.

Since a programmer has to specify the exact "how", it is easy to make a programming error. In addition, given the popularity of these programming languages, there are many bugs in software applications that are programmed in them. This motivates for a way to formally verify certain properties about programs written in these languages. Hoare logic is one way to mechanically reason about computer programs.

We will provide an implementation of an imperative language together with a toy implementation of Hoare logic that will allow us to reason about programs in this language. Our implementation of Hoare's logic will be bottom-up, in the sense that we will build the proofs from the ground up, in contrast to e.g. programming languages such as Dafny[1], that build proofs from top to bottom. That is, in Dafny, the user provides a proposition and Dafny will derive the proof, automatically, using the automated theorem prover Z3[2].

The programming language that we will implement already has an implementation in Coq[3]. However, implementing a language in Haskell is more concerned about playing at the value level (and to some extent at the type level), whereas in a dependently-typed language the focus is at the type level. Specifically for Hoare's logic, in Haskell we cannot use any of the meta language's constructs to do mathematical proofs, so we have to take care of these algorithms ourselves.

Haskell is not a strongly normalizing language, which means that not every evaluation necessarily terminates. However, in dependently-typed languages, such as Coq[4], the evaluation of proof terms (e.g. the type checker) is strongly normalizing, and this is what allows us to express mathematical proofs.

There are good introductory books on dependent types[3]. In some books, there is a gentler approach that might be handy for newcomers[5]. There are good introductions to Haskell as well[6].

2 Numbers language

We provide the syntax, evaluation rules, and the implementation in Haskell of a simple language that is strongly normalizing.

2.1 Arithmetic

We start by introducing a language that can do arithmetic. The syntax of the language expressed in BNF (Backus-Naur Form) is as follows:

```
digit  ::= "0" | "1" | ... | "8" | "9"
aexp   ::= aterm | aterm relop aterm
number ::= - digit* | digit*
relop  ::= "+" | "-" | "*"
aterm  ::= aexp | number | var
var    ::= A | B | C ... | Y | Z
```

Followed by a direct translation to Haskell code:

```
data Aexp =
  ANum Integer
  | AId Char
  | APlus Aexp Aexp
  | AMinus Aexp Aexp
  | AMult Aexp Aexp
```

We show the evaluation rules for optimizing an expression. Here, "optimization" is just another evaluation strategy.

$$\frac{}{\texttt{APlus}\ \$a_1\ \$a_2 \hookrightarrow \$(a_1 + a_2)}\ \text{(A-Opt-Plus)} \qquad \frac{}{\texttt{AMinus}\ \$a_1\ \$a_2 \hookrightarrow \$(a_1 - a_2)}\ \text{(A-Opt-Minus)}$$

$$\frac{}{\texttt{AMult}\ \$a_1\ \$a_2 \hookrightarrow \$(a_1 \cdot a_2)}\ \text{(A-Opt-Mult)}$$

That is, we lift the values from our language to the meta language (Haskell), where the symbol $ represents the value constructor `ANum`, and the symbol $\hookrightarrow$ represents the optimization function.

The same rules represented in Haskell:

```
aoptimize :: Aexp -> Aexp
aoptimize (APlus (ANum a1) (ANum a2)) = ANum (a1 + a2)
aoptimize (AMinus (ANum a1) (ANum a2)) = ANum (a1 - a2)
aoptimize (AMult (ANum a1) (ANum a2)) = ANum (a1 * a2)
aoptimize x = x
```

2.2 Boolean

Next, we introduce a language that can handle boolean expressions. The syntax of the language expressed in BNF is:

```
bexp   ::= bterm | bterm brelop bterm | aterm arelop aterm | unop bterm
arelop ::= "==" | "<"
brelop ::= "&&"
bterm  ::= "T" | "F"
unop   ::= "!"
```

The same syntax is represented with Haskell code:

```
data Bexp =
  BTrue
  | BFalse
  | BEq Aexp Aexp
  | BLe Aexp Aexp
```

```
  | BLt Aexp Aexp
  | BNot Bexp
  | BAnd Bexp Bexp
```

Similarly to `aoptimize`, we list the evaluation rules for optimizing a boolean expression:

$$\frac{a_1 = a_2}{\texttt{BEq}\ \$a_1\ \$a_2 \hookrightarrow \texttt{BTrue}}\ \text{(B-Opt-EqNumTrue)} \qquad \frac{a_1 \neq a_2}{\texttt{BEq}\ \$a_1\ \$a_2 \hookrightarrow \texttt{BFalse}}\ \text{(B-Opt-EqNumFalse)}$$

$$\frac{v_1 = v_2}{\texttt{BEq}\ \#v_1\ \#v_2 \hookrightarrow \texttt{BTrue}}\ \text{(B-Opt-EqIdTrue)} \qquad \frac{v_1 \neq v_2}{\texttt{BEq}\ \#v_1\ \#v_2 \hookrightarrow \texttt{BEq}\ \#v_1\ \#v_2}\ \text{(B-Opt-EqIdFalse)}$$

$$\frac{}{\texttt{BNot BTrue} \hookrightarrow \texttt{BFalse}}\ \text{(B-Opt-NegTrue)} \qquad \frac{}{\texttt{BNot BFalse} \hookrightarrow \texttt{BTrue}}\ \text{(B-Opt-NegFalse)}$$

$$\frac{}{\texttt{BAnd BFalse}\ x \hookrightarrow \texttt{BFalse}}\ \text{(B-Opt-AndFalse)} \qquad \frac{}{\texttt{BAnd BTrue}\ x \hookrightarrow x}\ \text{(B-Opt-AndTrue)}$$

Where the # symbol represents a variable (the value constructor `AId`).

These rules translate to the following implementation in Haskell:

```
boptimize :: Bexp -> Bexp
boptimize (BEq (ANum a1) (ANum a2)) = if a1 == a2 then BTrue else BFalse
boptimize (BEq (AId v1) (AId v2)) = if v1 == v2 then BTrue else BEq (AId v1) (AId v2)
boptimize (BNot BTrue) = BFalse
boptimize (BNot BFalse) = BTrue
boptimize (BAnd BFalse _) = BFalse
boptimize (BAnd BTrue b2) = b2
boptimize x = x
```

2.3 Evaluation

The optimization functions `aoptimize` and `boptimize` represent a simple single-step evaluation. Next, we introduce a way to do a full evaluation of arithmetic and boolean expressions.

To support variables, we introduce the notion of a context; one implementation in Haskell is simply a mapping from characters to numbers.

```
type Context = M.Map Char Integer
```

We list the evaluation rules for arithmetic expressions:

$$\frac{(\#v, \$v') \in ctx}{\#v \underset{ctx}{\rightarrow} \$v'}\ \text{(A-Eval-Id)} \qquad \frac{}{\$n \underset{ctx}{\rightarrow} n}\ \text{(A-Eval-Num)} \qquad \frac{a_1 \underset{ctx}{\rightarrow} a_1', a_2 \underset{ctx}{\rightarrow} a_2'}{\texttt{APlus}\ a_1\ a_2 \underset{ctx}{\rightarrow} a_1' + a_2'}\ \text{(A-Eval-Plus)}$$

$$\frac{a_1 \underset{ctx}{\rightarrow} a_1', a_2 \underset{ctx}{\rightarrow} a_2'}{\texttt{AMinus}\ a_1\ a_2 \underset{ctx}{\rightarrow} a_1' - a_2'}\ \text{(A-Eval-Minus)} \qquad \frac{a_1 \underset{ctx}{\rightarrow} a_1', a_2 \underset{ctx}{\rightarrow} a_2'}{\texttt{AMult}\ a_1\ a_2 \underset{ctx}{\rightarrow} a_1' \cdot a_2'}\ \text{(A-Eval-Mult)}$$

The arrow symbol $\underset{ctx}{\rightarrow}$ in the rules represents the actual evaluation of arithmetic expressions under context ctx.

Follows the implementation of these rules in Haskell:

```
aeval :: Context -> Aexp -> Integer
aeval ctx (AId v) = ctx M.! v -- element may not exist
aeval ctx (ANum n) = n
aeval ctx (APlus a1 a2) = aeval ctx a1 + aeval ctx a2
aeval ctx (AMinus a1 a2) = aeval ctx a1 - aeval ctx a2
aeval ctx (AMult a1 a2) = aeval ctx a1 * aeval ctx a2
```

Finally, the evaluation rules for boolean expressions:

$$\frac{}{\texttt{BTrue} \underset{ctx}{\Rightarrow} \texttt{True}}\ \text{(B-Eval-True)} \qquad \frac{}{\texttt{BFalse} \underset{ctx}{\Rightarrow} \texttt{False}}\ \text{(B-Eval-False)}$$

$$\frac{a_1 \underset{ctx}{\rightarrow} a_1', a_2 \underset{ctx}{\rightarrow} a_2'}{\texttt{BEq}\ a_1\ a_2 \underset{ctx}{\Rightarrow} a_1' = a_2'}\ \text{(B-Eval-Eq)} \qquad \frac{a_1 \underset{ctx}{\rightarrow} a_1', a_2 \underset{ctx}{\rightarrow} a_2'}{\texttt{BLe}\ a_1\ a_2 \underset{ctx}{\Rightarrow} a_1' \leq a_2'}\ \text{(B-Eval-Le)} \qquad \frac{a_1 \underset{ctx}{\rightarrow} a_1', a_2 \underset{ctx}{\rightarrow} a_2'}{\texttt{BLt}\ a_1\ a_2 \underset{ctx}{\Rightarrow} a_1' < a_2'}\ \text{(B-Eval-Lt)}$$

$$\frac{b \underset{ctx}{\Rightarrow} b'}{\texttt{BNot}\ b' \underset{ctx}{\Rightarrow} \neg b'}\ \text{(B-Eval-Not)} \qquad \frac{b_1 \underset{ctx}{\Rightarrow} b_1', b_2 \underset{ctx}{\Rightarrow} b_2'}{\texttt{BAnd}\ b_1\ b_2 \underset{ctx}{\Rightarrow} b_1' \wedge b_2'}\ \text{(B-Eval-And)}$$

Similarly, the double right arrow symbol $\underset{ctx}{\Rightarrow}$ in the rules represents the actual evaluation of boolean expressions under context ctx.

We provide the implementation of these rules in Haskell:

```
beval :: Context -> Bexp -> Bool
beval ctx BTrue = True
beval ctx BFalse = False
beval ctx (BEq a1 a2) = aeval ctx a1 == aeval ctx a2
beval ctx (BLe a1 a2) = aeval ctx a1 <= aeval ctx a2
beval ctx (BLt a1 a2) = aeval ctx a1 < aeval ctx a2
beval ctx (BNot b1) = not (beval ctx b1)
beval ctx (BAnd b1 b2) = beval ctx b1 && beval ctx b2
```

For example, the single-step optimization functions can be used as following:

```
> let e = BNot BTrue in "Optimize: " ++ show e ++ " = " ++ show (boptimize e)
Optimize: ! (TRUE) = FALSE
> let e = APlus (ANum 2) (ANum 5) in "Optimize: " ++ show e ++ " = " ++ show (aoptimize e)
Optimize: 2 + 5 = 7
```

However, since the single-step evaluation strategy does not rely on any context, it will not be able to deduce much about expressions that involve variables, e.g. `X + 5`, while `aeval` can substitute for this variable, given a context:

```
> let e = APlus (AId 'X') (ANum 5) in "Optimize: " ++ show e ++ " = " ++ show (aoptimize e)
Optimize: X + 5 = X + 5
> let e = APlus (AId 'X') (ANum 5) in show e ++ " = " ++ show (aeval (M.fromList [('X', 5)]) e)
X + 5 = 10
> let e = BEq (AId 'X') (ANum 5) in show e ++ " = " ++ show (beval (M.fromList [('X', 5)]) e)
X == 5 = True
```

3 Imperative language

We proceed with providing the syntax, evaluation rules, and the implementation in Haskell of an imperative language.

3.1 Commands

We show the data type of the imperative language expressed in Haskell:

```
data Command =
  CSkip
  | CAssign Char Aexp
  | CSequence Command Command
  | CIfElse Bexp Command Command
  | CWhile Bexp Command
  | CAssert Bexp Command Bexp
```

That is, the language contains the minimum set of commands which make an imperative language:

- `CSkip` is the the no operation command - the empty statement.
- `CAssign` will assign a value to a variable in a context.
- `CSequence` will join two commands, which allows for the evaluation of commands in sequence.
- `CIfElse` accepts a boolean and depending on its value either executes one command, or another.
- `CWhile` accepts a boolean and keeps executing a command as long as the boolean is true.
- `CAssert` accepts a precondition, a command, and a postcondition. The evaluation will be successful if the precondition and the postcondition are satisfied before and after executing the command, respectively.

We show the evaluation rules for this language:

$$\frac{}{\texttt{CSkip} \underset{ctx}{\mapsto} ctx} \text{(C-Eval-Skip)}$$

$$\frac{v \underset{ctx}{\rightarrow} v'}{\texttt{CAssign}\ c\ v \underset{ctx}{\mapsto} ctx \cup (c, v')} \text{(C-Eval-Assign)} \qquad \frac{c_1 \underset{ctx}{\mapsto} ctx', c_2 \underset{ctx'}{\mapsto} ctx''}{\texttt{CSequence}\ c_1\ c_2 \underset{ctx}{\mapsto} ctx''} \text{(C-Eval-Sequence)}$$

$$\frac{b \underset{ctx}{\Rightarrow} \texttt{True}, c_1 \underset{ctx}{\mapsto} ctx'}{\texttt{CIfElse}\ b\ c_1\ c_2 \underset{ctx}{\mapsto} ctx'} \text{(C-Eval-IfTrue)} \qquad \frac{b \underset{ctx}{\Rightarrow} \texttt{False}, c_2 \underset{ctx}{\mapsto} ctx'}{\texttt{CIfElse}\ b\ c_1\ c_2 \underset{ctx}{\mapsto} ctx'} \text{(C-Eval-IfFalse)}$$

$$\frac{b \underset{ctx}{\Rightarrow} \texttt{True}, c \underset{ctx}{\mapsto} ctx', \texttt{CWhile}\ b\ c \underset{ctx'}{\mapsto} ctx''}{\texttt{CWhile}\ b\ c \underset{ctx}{\mapsto} ctx''} \text{(C-Eval-WhileTrue)} \qquad \frac{b \underset{ctx}{\Rightarrow} \texttt{False}}{\texttt{CWhile}\ b\ c \underset{ctx}{\mapsto} ctx} \text{(C-Eval-WhileFalse)}$$

$$\frac{b_1 \underset{ctx}{\Rightarrow} \texttt{True}, c \underset{ctx}{\mapsto} ctx', b_2 \underset{ctx'}{\Rightarrow} \texttt{True}}{\texttt{CAssert}\ b_1\ c\ b_2} \text{(C-Eval-Assert)}$$

The map arrow symbol $\underset{ctx}{\mapsto}$ in the rules represents the actual evaluation of a command under context ctx.

In Coq, each implementation of the rules would be represented at the type level. However, since we're working at the value level in Haskell, we will rely on the `Either` data type to distinguish between provable and not provable terms.

```
eval :: Context -> Command -> Either String Context
eval ctx CSkip = Right ctx
eval ctx (CAssign c v) = Right $ M.insert c (aeval ctx v) ctx
eval ctx (CSequence c1 c2) = let ctx' = eval ctx c1 in whenRight ctx' (\ctx'' -> eval ctx'' c2)
eval ctx (CIfElse b c1 c2) = eval ctx $ if beval ctx b then c1 else c2
eval ctx (CWhile b c) =
  if beval ctx b
  then let ctx' = eval ctx c in whenRight ctx' (\ctx'' -> eval ctx'' (CWhile b c))
  else Right ctx
eval ctx (CAssert b1 c b2) =
  if beval ctx b1
  then whenRight (eval ctx c)
       (\ctx' -> if beval ctx' b2
                 then Right ctx'
                 else Left "Post-condition does not match!")
  else Left "Pre-condition does not match!"
```

Note that this language is not strongly normalizing; consider the evaluation of `CWhile BTrue CSkip`.

As an example, the factorial program (sequence of commands) can be represented with the pseudo-code:

```
Z := X
Y := 1
while (~Z = 0)
  Y := Y * Z
  Z := Z - 1
```

That is, this program will calculate `Y := X!`. Here's an implementation of it in our language:

```
> :{
| fact_X =
|   let l1 = CAssign 'Z' (AId 'X')
|       l2 = CAssign 'Y' (ANum 1)
|       l3 = CWhile (BNot (BEq (AId 'Z') (ANum 0))) (CSequence l4 l5)
|       l4 = CAssign 'Y' (AMult (AId 'Y') (AId 'Z'))
|       l5 = CAssign 'Z' (AMinus (AId 'Z') (ANum 1))
|   in CSequence l1 (CSequence l2 l3)
| :}
> eval (M.fromList [('X', 5)]) fact_X
Right (fromList [('X',5),('Y',120),('Z',0)])
> let e = CAssert (BEq (ANum 5) (AId 'X')) factX (BEq (ANum 120) (AId 'Y')) in "Assert {X=5} factX
    {Y=120}: " ++ show (eval (M.fromList [('X', 5)]) e)
Assert {X=5} factX {Y=120}: Right (fromList [('X',5),('Y',120),('Z',0)])
> let e = CAssert (BEq (ANum 4) (AId 'X')) factX (BEq (ANum 120) (AId 'Y')) in "Assert {X=4} factX
    {Y=120}: " ++ show (eval (M.fromList [('X', 5)]) e)
Assert {X=4} factX {Y=120}: Left "Pre-condition does not match!"
```

4 Hoare logic

In the previous chapter, we implemented assertions (`CAssert`) at the run-time (`eval`) level. The biggest disadvantage of that is we have to do a full evaluation to deduce some facts about programs; considering the assertion example of the `factX` program, it has to actually evaluate the factorial to conclude something. This motivates the need for an additional evaluation strategy that will allow us to deduce facts about programs without doing a full evaluation.

Some programming languages, like Python, don't have a compile step and the `eval` function we provided is kind of equivalent to evaluating programs in Python. But some programming languages do have a compile step, like C or Haskell, and this compilation step can be beneficial in that it can do additional different checks, e.g., type checks. That's what we'll do here - implement a "compile"-time check (just another evaluation strategy) using some of the rules in Hoare's logic, and this check can be used to check the validity of a program, before fully evaluating it.

We list the rules of Hoare logic, some of which are outlined in the original paper[7].

$$\frac{}{\{P\}\texttt{skip}\{P\}}\ \text{(H-Skip)} \qquad \frac{}{\{P[E/x]\}x := E\{P\}}\ \text{(H-Assign)}$$

$$\frac{P_1 \to P_2, \{P_2\}S\{Q_2\}, Q_2 \to Q_1}{\{P_1\}S\{Q_1\}}\ \text{(H-Consequence)} \qquad \frac{\{P\}S\{Q\}, \{Q\}T\{R\}}{\{P\}S;T\{R\}}\ \text{(H-Sequence)}$$

$$\frac{\{B \wedge P\}S\{Q\}, \{\neg B \wedge P\}T\{Q\}}{\{P\}\texttt{if } B \texttt{ then } S \texttt{ else } T\{Q\}}\ \text{(H-Conditional)} \qquad \frac{\{B \wedge P\}S\{P\}}{\{P\}\texttt{while } B \texttt{ do } S\{\neg B \wedge P\}}\ \text{(H-While)}$$

4.1 Implementation

We represent the Hoare triple as a product of a command, a precondition (`Bexp`), and a postcondition (`Bexp`). In this case, the design decision is that pre/postconditions will be in the language of `Bexp`, but they can also be in a different language (regardless of that `Bexp` is used in `Command`).

```
data HoareTriple = HoareTriple Bexp Command Bexp
```

Triples should not be constructed with `HoareTriple`, rather through the functions/rules that we provide next.

4.1.1 H-Skip

The Haskell implementation of the Hoare skip rule can be represented as follows:

```
hoareSkip :: Bexp -> HoareTriple
hoareSkip q = HoareTriple q CSkip q
```

Note that the validity of the precondition is not being checked (since that would require evaluating `Aexp`/`Bexp` with a context); the proof merely states that assuming some precondition, a command produces some postcondition.

```
> hoareSkip (BEq (ANum 3) (ANum 3))
{3 == 3} ; {3 == 3}
```

4.1.2 H-Assign

Let `Q[E/V]` denote the expression `Q` in which each free occurrence of `V` is replaced with `E`. Given an assignment command `V := E`, it should produce the triple where the precondition is `Q[E/V]` and the postcondition is `Q`, for any `Q`.

```
hoareAssignment :: Char -> Aexp -> Bexp -> HoareTriple
hoareAssignment v e q =
  HoareTriple
  (substBexp q (aoptimize e) v)
  (CAssign v e)
  q
```

There are several ways how the function `substBexp` can be implemented. We list a few:

- One way is to do a full `Aexp`/`Bexp` evaluation; we can do this, since these languages are strongly normalizing, compared to our imperative language. However, this evaluation can still take some time, and it requires a context in addition.
- Another way is to specify some concrete set of mathematical rewrites that can be applied, based on the original languages (`Aexp`/`Bexp`).

In this paper, the substitution will contain a concrete set of mathematical rewrites, though the reader is encouraged to try different implementations and spot the advantages and disadvantages of each.

```
substAexp :: Aexp -> Aexp -> Char -> Aexp
substAexp (AId x) e v = if x == v then e else AId x
substAexp (APlus x y) e v = APlus (substAexp x e v) (substAexp y e v)
substAexp (AMinus x y) e v = AMinus (substAexp x e v) (substAexp y e v)
substAexp (AMult x y) e v = AMult (substAexp x e v) (substAexp y e v)
substAexp x _ _ = x

substBexp :: Bexp -> Aexp -> Char -> Bexp
substBexp q@(BEq x y) e v = BEq (aoptimize $ substAexp x e v) (aoptimize $ substAexp y e v)
substBexp q@(BLe x y) e v = BLe (aoptimize $ substAexp x e v) (aoptimize $ substAexp y e v)
substBexp (BAnd b1 b2) e v = BAnd (substBexp b1 e v) (substBexp b2 e v)
substBexp (BNot b) e v = BNot (substBexp b e v)
substBexp q _ _ = q
```

This implementation will rewrite a variable with an arithmetical expression within a boolean expression. We can now deduce some proofs, such as:

```
> hoareAssignment 'X' (ANum 3) (BEq (AId 'X') (ANum 3))
{3 == 3} X := 3; {X == 3}
```

Given the precondition `3 = 3`, the command `X := 3` implies the postcondition `X = 3`.

4.1.3 H-Consequence

The consequence rule can be used to strengthen a precondition and/or weaken a postcondition in a Hoare triple.

Ideally, the implication in the consequence rule would represent the evaluation of a higher-order logic, but we're limiting it to optimization (our own small logic) for simplicity. That is, we use $\hookrightarrow$ in place of $\rightarrow$ in the rule.

In other words, the rule provides a way to transform a Hoare triple by embedding the result of an evaluation of a logic (in this case specifically, the optimization result of `boptimize`) into a Hoare triple.

```
hoareConsequence :: Bexp -> HoareTriple -> Bexp -> Either String HoareTriple
hoareConsequence p1 (HoareTriple p2 c q2) q1
  | boptimize p1 == p2 && q1 == q2 = Right $ HoareTriple p1 c q1
  | q1 == boptimize q2 && p1 == p2 = Right $ HoareTriple p1 c q1
```

```
hoareConsequence _ _ _ = Left "Cannot construct proof"
```

As with `eval`, the `Either` data type is used to distinguish between provable and not provable terms.

For example, considering the `hoareAssignment` example we saw earlier, we can modify its precondition as follows:

```
> hoareAssignment 'X' (ANum 3) (BEq (AId 'X') (ANum 3))
{3 == 3} X := 3; {X == 3}
> hoareConsequence (BAnd BTrue (BEq (ANum 3) (ANum 3))) (hoareAssignment 'X' (ANum 3) (BEq (AId
    'X') (ANum 3))) (BEq (AId 'X') (ANum 3))
Right {TRUE && 3 == 3} X := 3; {X == 3}
```

4.1.4 H-Sequence

For the Hoare sequence rule, given two Hoare triples, the postcondition of the first triple must be equivalent to the precondition of the second triple, for some definition of equivalent; in this specific case, we rely on Haskell's `Eq`.

```
hoareSequence :: HoareTriple -> HoareTriple -> Either String HoareTriple
hoareSequence (HoareTriple p c1 q1) (HoareTriple q2 c2 r)
  | q1 == q2 = Right $ HoareTriple p (CSequence c1 c2) r
hoareSequence _ _ = Left "Cannot construct proof"
```

Several commands can be chained as follows:

```
> let c1 = hoareAssignment 'Y' (ANum 1) (BAnd (BEq (AId 'Y') (ANum 1)) (BEq (AId 'X') (AId 'X')))
{1 == 1 && X == X} Y := 1; {Y == 1 && X == X}
> let c2 = hoareAssignment 'Z' (AId 'X') (BAnd (BEq (AId 'Y') (ANum 1)) (BEq (AId 'Z') (AId 'X')))
{Y == 1 && X == X} Z := X; {Y == 1 && Z == X}
> hoareSequence c1 c2
Right {1 == 1 && X == X} Y := 1; Z := X; {Y == 1 && Z == X}
```

4.1.5 H-Conditional

We provide an implementation for the Hoare conditional rule, covering both cases for commutativity of logical and.

```
hoareConditional :: HoareTriple -> HoareTriple -> Either String HoareTriple
hoareConditional (HoareTriple (BAnd b1 p1) c1 q1) (HoareTriple (BAnd (BNot b2) p2) c2 q2)
  | b1 == b2 &&
    p1 == p2 &&
    q1 == q2 = Right $ HoareTriple p1 (CIfElse b1 c1 c2) q1
hoareConditional (HoareTriple (BAnd p1 b1) c1 q1) (HoareTriple (BAnd (BNot p2) b2) c2 q2)
  | b1 == b2 &&
    p1 == p2 &&
    q1 == q2 = Right $ HoareTriple p1 (CIfElse b1 c1 c2) q1
hoareConditional _ _ = Left "Cannot construct proof"
```

For the purposes of example, we start by considering the command:

```
> CIfElse (BNot (BEq (AId 'X') (ANum 0))) CSkip (CAssign 'X' (APlus (AId 'X') (ANum 1)))
(If (! (X == 0)) Then (;) Else (X := X + 1;));
```

The conditionals `B` and `P` are given in the code. Now, following the rule to construct a Hoare triple for this command, we have to construct two proofs:

- $\{X \neq 0 \wedge P\}\,;\{Q\}$
- $\{\neg(X \neq 0) \wedge P\}X := X + 1;\{Q\}$

To be able to construct these proofs, we will add another substitution rule:

```
substBexp q@(BEq (AId x) (ANum 0)) (APlus (AId x2) (ANum y1)) v
  | x == x2 && x2 == v && y1 > 0 = BNot (BEq (AId x) (ANum 0))
```

```
  | otherwise = q
```

This substitution rule states that whenever x is zero, increasing it by a non-zero value will produce the postcondition that x is no longer zero. With this, we can construct the proof as follows:

```
> let eg1 = hoareSkip (BAnd (BNot (BEq (AId 'X') (ANum 0))) (BEq (ANum 0) (ANum 0)))
{! (X == 0) && 0 == 0} ; {! (X == 0) && 0 == 0}
> let eg2 = hoareAssignment 'X' (APlus (AId 'X') (ANum 1)) (BAnd (BNot (BEq (AId 'X') (ANum 0)))
    (BEq (ANum 0) (ANum 0)))
{! (! (X == 0)) && 0 == 0} X := X + 1; {! (X == 0) && 0 == 0}
> hoareConditional eg1 eg2
Right {0 == 0} (If (! (X == 0)) Then (;) Else (X := X + 1;)); {! (X == 0) && 0 == 0}
```

4.1.6 H-While

In this part, we provide the implementation of the last and most important and complex rule.

```
hoareWhile :: HoareTriple -> Either String HoareTriple
hoareWhile (HoareTriple (BAnd b p1) c p2)
  | p1 == p2 = Right $ HoareTriple p1 (CWhile b c) (BAnd (BNot b) p1)
hoareWhile (HoareTriple (BAnd p1 b) c p2)
  | p1 == p2 = Right $ HoareTriple p1 (CWhile b c) (BAnd (BNot b) p1)
hoareWhile _ = Left "Cannot construct proof"
```

The implementation for constructing the rule is straightforward, but the tricky part is that we have to properly construct the precondition and the postcondition so that they match the rule.

Similarly as we did with the `H-Conditional` rule, for the purposes of example, we will consider the command:

```
> CWhile BTrue CSkip
(While (TRUE) Do {;});
```

In this case, we get that `B` is `BTrue` and `S` is `CSkip`. Thus, we must provide the Hoare triple $\{\top \wedge P\}\ ;\{P\}$. We can use `hoareSkip` together with `hoareConsequence` to construct such triple as follows:

```
> let c = hoareSkip (BEq (ANum 0) (ANum 0))
{0 == 0} ; {0 == 0}
> hoareConsequence (BAnd BTrue (BEq (ANum 0) (ANum 0))) c (BEq (ANum 0) (ANum 0))
Right {TRUE && 0 == 0} ; {0 == 0}
```

Now we can deduce the following:

```
> whenRight (hoareConsequence (BAnd BTrue (BEq (ANum 0) (ANum 0))) c (BEq (ANum 0) (ANum 0))) (\x
    -> hoareWhile x)
Right {0 == 0} (While (TRUE) Do {;}); {! (TRUE) && 0 == 0}
```

That is, from an infinite loop ($\bot$), follows anything (false).

4.2 Example proofs

So far we only looked at artificial examples, neatly crafted to match the specific forms that the rules accept. Even though the implementation we provided is simple enough, it can still be used to prove some useful real-world facts.

4.2.1 Assignment (Swap)

We will show a simple example of the assignment rule.

```
> hoareAssignment 'a' (APlus (AId 'a') (AId 'b')) (BAnd (BEq (AMinus (AId 'a') (AId 'b')) (AId
    'A')) (BEq (AId 'b') (AId 'B')))
{(a + b - b) == A && b == B} a := a + b; {(a - b) == A && b == B}
```

```
> hoareAssignment 'b' (AMinus (AId 'a') (AId 'b')) (BAnd (BEq (AId 'b') (AId 'A')) (BEq (AMinus
    (AId 'a') (AId 'b')) (AId 'B')))
{(a - b) == A && (a - (a - b)) == B} b := (a - b); {b == A && (a - b) == B}
> hoareAssignment 'a' (AMinus (AId 'a') (AId 'b')) (BAnd (BEq (AId 'b') (AId 'A')) (BEq (AId 'a')
    (AId 'B')))
{b == A && (a - b) == B} a := (a - b); {b == A && a == B}
```

That is, we start with the precondition that `a == A && b == B` and we reach `b == A && a == B`. We proved that this set of commands will swap the values between two variables. To make this proof more obvious, we can add two additional optimization rules that state $a + b - b = a$ and $a - (a - b) = b$:

```
aoptimize q@(AMinus (APlus (AId a1) (AId a2)) (AId a3)) = if a2 == a3 then AId a1 else q
aoptimize q@(AMinus (AId a1) (AMinus (AId a2) (AId a3))) = if a1 == a2 then AId a3 else q
```

The proof can be completed with `hoareSequence` as follows:

```
> whenRight (hoareSequence swap1 swap2) (\x -> hoareSequence x swap3)
Right {a == A && b == B} a := a + b; b := (a - b); a := (a - b); {b == A && a == B}
```

4.2.2 While (Counting)

We provide the implementation of a program that increases `X` up to 10:

```
> :{
| countTo10 =
|   let l1 = CAssign 'X' $ ANum 0
|       l2 = CWhile (BLt (AId 'X') (ANum 10)) l3
|       l3 = CAssign 'X' (APlus (AId 'X') (ANum 1))
|   in CSequence l1 l2
> eval M.empty countTo10
Right (fromList [('X',10)])
```

Looking at the while rule, we can determine the values of `B` and `S` based on the command:

$$\frac{\{x < 10 \wedge P\}\ \texttt{x := x + 1}\ \{P\}}{\{P\}\texttt{while}\ x < 10\ \texttt{do x := x + 1}\ \{\neg(x < 10) \wedge P\}}$$

A sensible value for `P` would be $x \leq 10$, because it holds wherever `P` is referenced in the formula.

$$\frac{\{x < 10 \wedge x \leq 10\}\ \texttt{x := x + 1}\ \{x \leq 10\}}{\{x \leq 10\}\texttt{while}\ x < 10\ \texttt{do x := x + 1}\ \{\neg(x < 10) \wedge x \leq 10\}}$$

Note that $\neg(x < 10) \wedge x \leq 10$ simplifies to $x = 10$.

We should not use `HoareTriple` directly, but it can still be useful to show us what we need to prove:

```
> let ht = HoareTriple (BAnd (BLt (AId 'X') (ANum 10)) (BLe (AId 'X') (ANum 10))) (CAssign 'X'
    (APlus (AId 'X') (ANum 1))) (BLe (AId 'X') (ANum 10))
{X < 10 && X <= 10} X := X + 1; {X <= 10}
> hoareWhile ht
Right {X <= 10} (While (X < 10) Do {X := X + 1;}); {! (X < 10) && X <= 10}
```

If we are able to produce `ht` using the rules, we'll be able to prove our statement. Using `hoareAssignment`, we can produce the following:

```
> let ht = hoareAssignment 'X' (APlus (AId 'X') (ANum 1)) (BLe (AId 'X') (ANum 10))
{X + 1 <= 10} X := X + 1; {X <= 10}
```

If we tweak our optimization and substitution rules, we will be able to show that we can get from `X + 1 <= 10` to `X <= 9`, and then to `X < 10`, using the `H-Consequence` rule.

5 Conclusion and future work

Compile-time, run-time, etc. are all about having evaluations at different levels. There is still computation going on, but the computation strategies at the compile-time level may be different from those at the run-time level. A full evaluation of `Command` can be expensive, and sometimes even not terminate, and we wanted a way to deduce propositions without doing a full evaluation. We showed how to achieve this, by providing a simple implementation of Hoare logic as a proof of concept.

The evaluation strategies of a programming language directly affect how a programmer thinks. For example, if a programming language has a type system, the programmer will take advantage of it, subconsciously being aware of the (different) evaluation strategies for the type checker, in addition to the regular evaluation strategy for programs. If a programming language has the feature to verify properties about programs, such as Hoare logic, the programmer will be aware of this different evaluation strategy and use it to write more correct software.

Even though the mathematical formulas for Hoare logic look simple, implementing them is a different matter. The implementation details cover stuff like ”what expressions do we want to support”, ”are we working with a strongly normalizing language”, ”what’s the language that will represent propositions”, etc. while these details are hidden in the mathematical representation of these formulas.

The biggest disadvantage of the provided implementation is that we had to manually add optimization rules, which is a bit tedious. Ideally, we would rely on a theorem prover that only contains a minimal set of axioms, and uses those to derive e.g. $x+1 \leq 10 \rightarrow x < 10$. Another disadvantage is that the implementation of `H-Consequence` is very simple, not really supporting logical implication as such. Finally, our system does not support quantifiers. Nevertheless, this paper serves as a good way to show the complexity and difficulty of designing such systems.

6 Conflict of interest

The author declares that they have no conflict of interest.

References

[1] Rustan M. Leino Dafny: An automatic program verifier for functional correctness *International Conference on Logic for Programming Artificial Intelligence and Reasoning, pp. 348-370. Springer, Berlin, Heidelberg*, 2010.

[2] Leonardo De Moura, Nikolaj Bjørner Z3: An efficient SMT solver. *International conference on Tools and Algorithms for the Construction and Analysis of Systems. Springer, Berlin, Heidelberg*, 2008.

[3] Benjamin C. Pierce, Arthur Azevedo de Amorim, Chris Casinghino, Marco Gaboardi, Michael Greenberg, Cătălin Hriţcu, Vilhelm Sjöberg, Brent Yorgey Logical Foundations *Electronic textbook*, 2020.

[4] Bruno Barras, Samuel Boutin, Cristina Cornes, Judicaël Courant, Jean-Christophe Filliâtre, et al. The Coq Proof Assistant Reference Manual: Version 6.1 *[Research Report] RT-0203, INRIA*, 1997.

[5] Boro Sitnikovski Gentle Introduction to Dependent Types with Idris *Leanpub/Amazon KDP*, 2018.

[6] Miran Lipovaca Learn You a Haskell For Great Good *No Starch Press*, 2011.

[7] Charles A.R. Hoare An axiomatic basis for computer programming *Communications of the ACM, 12(10), pp.576-580*, 1969.